Rules for integrands of the form $(a + b \operatorname{Sin}[e + f x])^m (c + d \operatorname{Sin}[e + f x])^n (A + B \operatorname{Sin}[e + f x])$

1: $\int \operatorname{Sin}[e + f x]^n (a + b \operatorname{Sin}[e + f x])^m (A + B \operatorname{Sin}[e + f x]) \, dx$ when $A b + a B == 0 \land a^2 - b^2 == 0 \land m \in \mathbb{Z} \land n \in \mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $A b + a B == 0 \land a^2 - b^2 == 0 \land m \in \mathbb{Z} \land n \in \mathbb{Z}$, then

$$\int \operatorname{Sin}[e + f x]^n (a + b \operatorname{Sin}[e + f x])^m (A + B \operatorname{Sin}[e + f x]) \, dx \to \int \operatorname{ExpandTrig}\left[\operatorname{Sin}[e + f x]^n (a + b \operatorname{Sin}[e + f x])^m (A + B \operatorname{Sin}[e + f x]), x\right] dx$$

Program code:

```
Int[sin[e_.+f_.*x_]^n_.*(a_+b_.*sin[e_.+f_.*x_])^m_.*(A_.+B_.*sin[e_.+f_.*x_]),x_Symbol] :=
  Int[ExpandTrig[sin[e+f*x]^n*(a+b*sin[e+f*x])^m*(A+B*sin[e+f*x]),x],x] /;
FreeQ[{a,b,e,f,A,B},x] && EqQ[A*b+a*B,0] && EqQ[a^2-b^2,0] && IntegerQ[m] && IntegerQ[n]
```

2: $\int (a + b \operatorname{Sin}[e + f x])^m (c + d \operatorname{Sin}[e + f x])^n (A + B \operatorname{Sin}[e + f x]) \, dx$ when $b c + a d == 0 \land a^2 - b^2 == 0 \land m \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b c + a d == 0 \land a^2 - b^2 == 0$, then $(a + b \operatorname{Sin}[z]) (c + d \operatorname{Sin}[z]) == a c \operatorname{Cos}[z]^2$

Rule: If $b c + a d == 0 \land a^2 - b^2 == 0 \land m \in \mathbb{Z}$, then

$$\int (a + b \operatorname{Sin}[e + f x])^m (c + d \operatorname{Sin}[e + f x])^n (A + B \operatorname{Sin}[e + f x]) \, dx \to a^m c^m \int \operatorname{Cos}[e + f x]^{2 m} (c + d \operatorname{Sin}[e + f x])^{n-m} (A + B \operatorname{Sin}[e + f x]) \, dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_+d_.*sin[e_.+f_.*x_])^n_.*(A_.+B_.*sin[e_.+f_.*x_]),x_Symbol] :=
  a^m*c^m*Int[Cos[e+f*x]^(2*m)*(c+d*Sin[e+f*x])^(n-m)*(A+B*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,A,B,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IntegerQ[m] && Not[IntegerQ[n] && (LtQ[m,0] && GtQ[n,0] || LtQ[0,n,m] || L
```

3: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])\,(A+B\,\text{Sin}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0$

Derivation: Algebraic expansion

Rule: If $b\,c-a\,d\neq 0$, then

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])\,(A+B\,\text{Sin}[e+f\,x])\,dx \longrightarrow \int (a+b\,\text{Sin}[e+f\,x])^m\,\left(A\,c+(B\,c+A\,d)\,\text{Sin}[e+f\,x]+B\,d\,\text{Sin}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_.*(c_.+d_.*sin[e_.+f_.*x_])*(A_.+B_.*sin[e_.+f_.*x_]),x_Symbol] :=
  Int[(a+b*Sin[e+f*x])^m*(A*c+(B*c+A*d)*Sin[e+f*x]+B*d*Sin[e+f*x]^2),x] /;
FreeQ[{a,b,c,d,e,f,A,B,m},x] && NeQ[b*c-a*d,0]
```

4. $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx$ when $b\,c+a\,d==0\,\wedge\,a^2-b^2==0\,\wedge\,m\notin\mathbb{Z}\,\wedge\,n\notin\mathbb{Z}$

   1. $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx$ when $b\,c+a\,d==0\,\wedge\,a^2-b^2==0\,\wedge\,m\notin\mathbb{Z}\,\wedge\,n\notin\mathbb{Z}\,\wedge\,A\,b\,(m+n+1)+a\,B\,(m-n)==0$

      1: $\int \frac{A+B\,\text{Sin}[e+f\,x]}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$ when $b\,c+a\,d==0\,\wedge\,a^2-b^2==0$

      Derivation: Algebraic expansion

      Basis: If $b\,c+a\,d==0\,\wedge\,a^2-b^2==0$, then $b\,c+a\,d==0$

      Basis: If $b\,c+a\,d==0$, then $\frac{A+B\,z}{\sqrt{a+b\,z}\,\sqrt{c+d\,z}}==\frac{(A\,b+a\,B)\,\sqrt{a+b\,z}}{2\,a\,b\,\sqrt{c+d\,z}}+\frac{(B\,c+A\,d)\,\sqrt{c+d\,z}}{2\,c\,d\,\sqrt{a+b\,z}}$

      Rule: If $b\,c+a\,d==0\,\wedge\,a^2-b^2==0$, then

$$\int \frac{A+B\,\text{Sin}[e+f\,x]}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx\;\rightarrow\;\frac{A\,b+a\,B}{2\,a\,b}\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx+\frac{B\,c+A\,d}{2\,c\,d}\int \frac{\sqrt{c+d\,\text{Sin}[e+f\,x]}}{\sqrt{a+b\,\text{Sin}[e+f\,x]}}\,dx$$

      Program code:

```
Int[(A_.+B_.*sin[e_.+f_.*x_])/(Sqrt[a_+b_.*sin[e_.+f_.*x_]]*Sqrt[c_+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  (A*b+a*B)/(2*a*b)*Int[Sqrt[a+b*Sin[e+f*x]]/Sqrt[c+d*Sin[e+f*x]],x] +
  (B*c+A*d)/(2*c*d)*Int[Sqrt[c+d*Sin[e+f*x]]/Sqrt[a+b*Sin[e+f*x]],x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0]
```

2: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,(A+B\,\mathrm{Sin}[e+f\,x])\,dx$ when $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m\notin\mathbb{Z}\,\land\,n\notin\mathbb{Z}\,\land\,A\,b\,(m+n+1)+a\,B\,(m-n)==0\,\land\,m\neq-\frac{1}{2}$

Derivation: Algebraic expansion and doubly degenerate sine recurrence 1c with $p\to 0$ and $A\,b\,(m+n+1)+a\,B\,(m-n)==0$

Basis: $A+B\,z==\frac{A\,b-a\,B}{b}+\frac{B\,(a+b\,z)}{b}$

Rule: If $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,A\,b\,(m+n+1)+a\,B\,(m-n)==0\,\land\,m\notin\mathbb{Z}\,\land\,m\neq-\frac{1}{2}$, then

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,(A+B\,\mathrm{Sin}[e+f\,x])\,dx\,\longrightarrow\,-\frac{B\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n}{f\,(m+n+1)}$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_.*(A_.+B_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -B*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n/(f*(m+n+1)) /;
FreeQ[{a,b,c,d,e,f,A,B,m,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && EqQ[A*b*(m+n+1)+a*B*(m-n),0] && NeQ[m,-1/2]
```

2: $\int\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\left(c+d\,\text{Sin}[e+f\,x]\right)^n\left(A+B\,\text{Sin}[e+f\,x]\right)dx$ when $b\,c+a\,d==0\,\land\,a^2-b^2==0$

Derivation: Algebraic expansion

Baisi: $A+B\,z==\frac{B\,(c+d\,z)}{d}-\frac{B\,c-A\,d}{d}$

Rule: If $b\,c+a\,d==0\,\land\,a^2-b^2==0$, then

$$\int\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\left(c+d\,\text{Sin}[e+f\,x]\right)^n\left(A+B\,\text{Sin}[e+f\,x]\right)dx\longrightarrow$$
$$\frac{B}{d}\int\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\left(c+d\,\text{Sin}[e+f\,x]\right)^{n+1}dx-\frac{B\,c-A\,d}{d}\int\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\left(c+d\,\text{Sin}[e+f\,x]\right)^n dx$$

Program code:

```
Int[Sqrt[a_.+b_.*sin[e_.+f_.*x_]]*(c_+d_.*sin[e_.+f_.*x_])^n_*(A_.+B_.*sin[e_.+f_.*x_]),x_Symbol] :=
  B/d*Int[Sqrt[a+b*Sin[e+f*x]]*(c+d*Sin[e+f*x])^(n+1),x] -
  (B*c-A*d)/d*Int[Sqrt[a+b*Sin[e+f*x]]*(c+d*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,A,B,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0]
```

3: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx$ when $b\,c+a\,d==0\,\wedge\,a^2-b^2==0\,\wedge\,m<-\frac{1}{2}$

Derivation: Algebraic expansion and doubly degenerate sine recurrence 1c with $p\to 0$

Basis: $A+B\,z==\frac{A\,b-a\,B}{b}+\frac{B\,(a+b\,z)}{b}$

Rule: If $b\,c+a\,d==0\,\wedge\,a^2-b^2==0\,\wedge\,m<-\frac{1}{2}$, then

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx\,\longrightarrow$$

$$\frac{(A\,b-a\,B)\,\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n}{a\,f\,(2\,m+1)}+\frac{a\,B\,(m-n)+A\,b\,(m+n+1)}{a\,b\,(2\,m+1)}\int (a+b\,\text{Sin}[e+f\,x])^{m+1}\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_.*(A_.+B_.*sin[e_.+f_.*x_]),x_Symbol] :=
  (A*b-a*B)*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n/(a*f*(2*m+1)) +
  (a*B*(m-n)+A*b*(m+n+1))/(a*b*(2*m+1))*Int[(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,A,B,m,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && (LtQ[m,-1/2] || ILtQ[m+n,0] && Not[SumSimplerQ[n,1]]) && NeQ[2*m+1,0]
```

4: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,(A+B\,\mathrm{Sin}[e+f\,x])\,dx$ when $b\,c+a\,d=0\,\wedge\,a^2-b^2=0\,\wedge\,m\nless-\frac{1}{2}$

Derivation: Algebraic expansion and doubly degenerate sine recurrence 1b with $m\to m+1,\ p\to 0$

Basis: $A+B\,z=\frac{A\,b-a\,B}{b}+\frac{B\,(a+b\,z)}{b}$

Rule: If $b\,c+a\,d=0\,\wedge\,a^2-b^2=0\,\wedge\,m\nless-\frac{1}{2}$, then

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,(A+B\,\mathrm{Sin}[e+f\,x])\,dx\,\longrightarrow$$

$$-\frac{B\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n}{f\,(m+n+1)}-\frac{B\,c\,(m-n)-A\,d\,(m+n+1)}{d\,(m+n+1)}\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_+d_.*sin[e_.+f_.*x_])^n_*(A_.+B_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -B*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n/(f*(m+n+1)) -
  (B*c*(m-n)-A*d*(m+n+1))/(d*(m+n+1))*Int[(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,A,B,m,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && Not[LtQ[m,-1/2]] && NeQ[m+n+1,0]
```

5. $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,(A+B\,\mathrm{Sin}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2=0\,\wedge\,c^2-d^2\neq 0$

1: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,(A+B\,\mathrm{Sin}[e+f\,x])$
dx when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2=0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m+n+2=0\,\wedge\,A\,(a\,d\,m+b\,c\,(n+1))-B\,(a\,c\,m+b\,d\,(n+1))=0$

Rule: If
$b\,c-a\,d\neq 0\,\wedge\,a^2-b^2=0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m+n+2=0\,\wedge\,A\,(a\,d\,m+b\,c\,(n+1))-B\,(a\,c\,m+b\,d\,(n+1))=0$,
then

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,(A+B\,\mathrm{Sin}[e+f\,x])\,dx\,\longrightarrow$$

$$\frac{(B\,c-A\,d)\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^{n+1}}{f\,(n+1)\,\left(c^2-d^2\right)}$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_*(A_.+B_.*sin[e_.+f_.*x_]),x_Symbol] :=
  (B*c-A*d)*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^(n+1)/(f*(n+1)*(c^2-d^2)) /;
FreeQ[{a,b,c,d,e,f,A,B,m,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && EqQ[m+n+2,0] && EqQ[A*(a*d*m+b*c*(n+1))-B*(a*c*m+b*d*
```

2. $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,(A+B\,\mathrm{Sin}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,m>\frac{1}{2}$

   1: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,(A+B\,\mathrm{Sin}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,m>\frac{1}{2}\,\land\,n<-1$

Derivation: Singly degenerate sine recurrence 1a with $p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,m>\frac{1}{2}\,\land\,n<-1$, then

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,(A+B\,\mathrm{Sin}[e+f\,x])\,dx\longrightarrow$$

$$-\frac{b^2\,(B\,c-A\,d)\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^{m-1}\,(c+d\,\mathrm{Sin}[e+f\,x])^{n+1}}{d\,f\,(n+1)\,(b\,c+a\,d)}-$$

$$\frac{b}{d\,(n+1)\,(b\,c+a\,d)}\int (a+b\,\mathrm{Sin}[e+f\,x])^{m-1}\,(c+d\,\mathrm{Sin}[e+f\,x])^{n+1}\cdot$$

$$(a\,A\,d\,(m-n-2)-B\,(a\,c\,(m-1)+b\,d\,(n+1))-(A\,b\,d\,(m+n+1)-B\,(b\,c\,m-a\,d\,(n+1)))\,\mathrm{Sin}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_*(A_.+B_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -b^2*(B*c-A*d)*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^(n+1)/(d*f*(n+1)*(b*c+a*d)) -
  b/(d*(n+1)*(b*c+a*d))*Int[(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^(n+1)*
    Simp[a*A*d*(m-n-2)-B*(a*c*(m-1)+b*d*(n+1))-(A*b*d*(m+n+1)-B*(b*c*m-a*d*(n+1)))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && GtQ[m,1/2] && LtQ[n,-1] &&
  IntegerQ[2*m] && (IntegerQ[2*n] || EqQ[c,0])
```

2: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m>\frac{1}{2}\,\wedge\,n\not<-1$

Derivation: Singly degenerate sine recurrence 1b with $p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m>\frac{1}{2}\,\wedge\,n\not<-1$, then

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx\,\longrightarrow$$
$$-\frac{b\,B\,\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^{m-1}\,(c+d\,\text{Sin}[e+f\,x])^{n+1}}{d\,f\,(m+n+1)}+$$
$$\frac{1}{d\,(m+n+1)}\int (a+b\,\text{Sin}[e+f\,x])^{m-1}\,(c+d\,\text{Sin}[e+f\,x])^n\cdot$$
$$(a\,A\,d\,(m+n+1)+B\,(a\,c\,(m-1)+b\,d\,(n+1))+(A\,b\,d\,(m+n+1)-B\,(b\,c\,m-a\,d\,(2\,m+n)))\,\text{Sin}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_*(A_.+B_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -b*B*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^(n+1)/(d*f*(m+n+1)) +
  1/(d*(m+n+1))*Int[(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^n*
    Simp[a*A*d*(m+n+1)+B*(a*c*(m-1)+b*d*(n+1))+(A*b*d*(m+n+1)-B*(b*c*m-a*d*(2*m+n)))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && GtQ[m,1/2] && Not[LtQ[n,-1]] && IntegerQ[2*m] &&
  (IntegerQ[2*n] || EqQ[c,0])
```

3. $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m<-\frac{1}{2}$

1: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m<-\frac{1}{2}\,\wedge\,n>0$

Derivation: Singly degenerate sine recurrence 2a with $p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m<-\frac{1}{2}\,\wedge\,n>0$, then

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx\,\longrightarrow$$

$$\frac{(A\,b-a\,B)\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n}{a\,f\,(2\,m+1)}-$$
$$\frac{1}{a\,b\,(2\,m+1)}\int(a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Sin}[e+f\,x])^{n-1}\cdot$$
$$(A\,(a\,d\,n-b\,c\,(m+1))-B\,(a\,c\,m+b\,d\,n)-d\,(a\,B\,(m-n)+A\,b\,(m+n+1))\,\mathrm{Sin}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_*(A_.+B_.*sin[e_.+f_.*x_]),x_Symbol] :=
  (A*b-a*B)*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n/(a*f*(2*m+1)) -
  1/(a*b*(2*m+1))*Int[(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^(n-1)*
    Simp[A*(a*d*n-b*c*(m+1))-B*(a*c*m+b*d*n)-d*(a*B*(m-n)+A*b*(m+n+1))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && LtQ[m,-1/2] && GtQ[n,0] && IntegerQ[2*m] &&
  (IntegerQ[2*n] || EqQ[c,0])
```

2: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,m<-\frac{1}{2}\,\land\,n\ngtr 0$

Derivation: Singly degenerate sine recurrence 2b with $p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,m<-\frac{1}{2}\,\land\,n\ngtr 0$, then

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx\,\longrightarrow$$
$$\frac{b\,(A\,b-a\,B)\,\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^{n+1}}{a\,f\,(2\,m+1)\,(b\,c-a\,d)}+$$
$$\frac{1}{a\,(2\,m+1)\,(b\,c-a\,d)}\int (a+b\,\text{Sin}[e+f\,x])^{m+1}\,(c+d\,\text{Sin}[e+f\,x])^n\,\cdot$$
$$(B\,(a\,c\,m+b\,d\,(n+1))+A\,(b\,c\,(m+1)-a\,d\,(2\,m+n+2))+d\,(A\,b-a\,B)\,(m+n+2)\,\text{Sin}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_*(A_.+B_.*sin[e_.+f_.*x_]),x_Symbol] :=
  b*(A*b-a*B)*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^(n+1)/(a*f*(2*m+1)*(b*c-a*d)) +
  1/(a*(2*m+1)*(b*c-a*d))*Int[(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^n*
    Simp[B*(a*c*m+b*d*(n+1))+A*(b*c*(m+1)-a*d*(2*m+n+2))+d*(A*b-a*B)*(m+n+2)*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && LtQ[m,-1/2] && Not[GtQ[n,0]] && IntegerQ[2*m] &&
  (IntegerQ[2*n] || EqQ[c,0])
```

4. $\int \sqrt{a+b\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0$

1: $\int \sqrt{a+b\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,A\,b\,d\,(2\,n+3)-B\,(b\,c-2\,a\,d\,(n+1))==0$

Derivation: Singly degenerate sine recurrence 1a with $B\to-\frac{A\,b\,(3+2\,n)}{2\,a\,(1+n)}$, $m\to\frac{1}{2}$, $p\to 0$

Derivation: Singly degenerate sine recurrence 1b with $B\to-\frac{A\,b\,(3+2\,n)}{2\,a\,(1+n)}$, $m\to\frac{1}{2}$, $p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,A\,b\,d\,(2\,n+3)-B\,(b\,c-2\,a\,d\,(n+1))==0$, then

$$\int \sqrt{a+b\,\text{Sin}[e+f\,x]}\,\left(c+d\,\text{Sin}[e+f\,x]\right)^n\,\left(A+B\,\text{Sin}[e+f\,x]\right)\,dx \longrightarrow -\frac{2\,b\,B\,\text{Cos}[e+f\,x]\,\left(c+d\,\text{Sin}[e+f\,x]\right)^{n+1}}{d\,f\,(2\,n+3)\,\sqrt{a+b\,\text{Sin}[e+f\,x]}}$$

Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]*(c_.+d_.*sin[e_.+f_.*x_])^n_*(A_.+B_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -2*b*B*Cos[e+f*x]*(c+d*Sin[e+f*x])^(n+1)/(d*f*(2*n+3)*Sqrt[a+b*Sin[e+f*x]]) /;
FreeQ[{a,b,c,d,e,f,A,B,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && EqQ[A*b*d*(2*n+3)-B*(b*c-2*a*d*(n+1)),0]
```

2: $\int \sqrt{a+b\,\text{Sin}[e+f\,x]}\,\left(c+d\,\text{Sin}[e+f\,x]\right)^n\,\left(A+B\,\text{Sin}[e+f\,x]\right)\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,n<-1$

Derivation: Singly degenerate sine recurrence 1a with $m\to\frac{1}{2},\ p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,n<-1$, then

$$\int \sqrt{a+b\,\text{Sin}[e+f\,x]}\,\left(c+d\,\text{Sin}[e+f\,x]\right)^n\,\left(A+B\,\text{Sin}[e+f\,x]\right)\,dx \longrightarrow$$

$$-\frac{b^2\,(B\,c-A\,d)\,\text{Cos}[e+f\,x]\,\left(c+d\,\text{Sin}[e+f\,x]\right)^{n+1}}{d\,f\,(n+1)\,(b\,c+a\,d)\,\sqrt{a+b\,\text{Sin}[e+f\,x]}}+$$
$$\frac{A\,b\,d\,(2\,n+3)-B\,(b\,c-2\,a\,d\,(n+1))}{2\,d\,(n+1)\,(b\,c+a\,d)}\int \sqrt{a+b\,\text{Sin}[e+f\,x]}\,\left(c+d\,\text{Sin}[e+f\,x]\right)^{n+1}\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]*(c_.+d_.*sin[e_.+f_.*x_])^n_*(A_.+B_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -b^2*(B*c-A*d)*Cos[e+f*x]*(c+d*Sin[e+f*x])^(n+1)/(d*f*(n+1)*(b*c+a*d)*Sqrt[a+b*Sin[e+f*x]]) +
  (A*b*d*(2*n+3)-B*(b*c-2*a*d*(n+1)))/(2*d*(n+1)*(b*c+a*d))*Int[Sqrt[a+b*Sin[e+f*x]]*(c+d*Sin[e+f*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && LtQ[n,-1]
```

3: $\int \sqrt{a+b\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0 \land a^2-b^2==0 \land c^2-d^2\neq 0 \land n \nless -1$

Derivation: Singly degenerate sine recurrence 1b with $m \to \frac{1}{2}$, $p \to 0$

Rule: If $b\,c-a\,d\neq 0 \land a^2-b^2==0 \land c^2-d^2\neq 0 \land n \nless -1$, then

$$\int \sqrt{a+b\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx \longrightarrow$$

$$-\frac{2\,b\,B\,\text{Cos}[e+f\,x]\,(c+d\,\text{Sin}[e+f\,x])^{n+1}}{d\,f\,(2\,n+3)\,\sqrt{a+b\,\text{Sin}[e+f\,x]}}+$$

$$\frac{A\,b\,d\,(2\,n+3)-B\,(b\,c-2\,a\,d\,(n+1))}{b\,d\,(2\,n+3)}\int \sqrt{a+b\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]*(c_.+d_.*sin[e_.+f_.*x_])^n_*(A_.+B_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -2*b*B*Cos[e+f*x]*(c+d*Sin[e+f*x])^(n+1)/(d*f*(2*n+3)*Sqrt[a+b*Sin[e+f*x]]) +
  (A*b*d*(2*n+3)-B*(b*c-2*a*d*(n+1)))/(b*d*(2*n+3))*Int[Sqrt[a+b*Sin[e+f*x]]*(c+d*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,A,B,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && Not[LtQ[n,-1]]
```

5: $\int \frac{A+B\,\text{Sin}[e+f\,x]}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0 \land a^2-b^2==0 \land c^2-d^2\neq 0$

Derivation: Algebraic expansion

Baisi: $A+B\,z == \frac{A\,b-a\,B}{b}+\frac{B\,(a+b\,z)}{b}$

Rule: If $b\,c-a\,d\neq 0 \land a^2-b^2==0 \land c^2-d^2\neq 0$, then

$$\int \frac{A+B\,\mathrm{Sin}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx \to \frac{A\,b-a\,B}{b}\int \frac{1}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx+\frac{B}{b}\int \frac{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}{\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx$$

Program code:

```
Int[(A_.+B_.*sin[e_.+f_.*x_])/(Sqrt[a_+b_.*sin[e_.+f_.*x_]]*Sqrt[c_.+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  (A*b-a*B)/b*Int[1/(Sqrt[a+b*Sin[e+f*x]]*Sqrt[c+d*Sin[e+f*x]]),x] +
  B/b*Int[Sqrt[a+b*Sin[e+f*x]]/Sqrt[c+d*Sin[e+f*x]],x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

6: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,(A+B\,\mathrm{Sin}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,n>0$

Derivation: Singly degenerate sine recurrence 2c with $p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,n>0$, then

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,(A+B\,\mathrm{Sin}[e+f\,x])\,dx\to$$
$$-\frac{B\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n}{f\,(m+n+1)}+$$
$$\frac{1}{b\,(m+n+1)}\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^{n-1}\,(A\,b\,c\,(m+n+1)+B\,(a\,c\,m+b\,d\,n)+(A\,b\,d\,(m+n+1)+B\,(a\,d\,m+b\,c\,n))\,\mathrm{Sin}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_*(A_.+B_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -B*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n/(f*(m+n+1)) +
  1/(b*(m+n+1))*Int[(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^(n-1)*
    Simp[A*b*c*(m+n+1)+B*(a*c*m+b*d*n)+(A*b*d*(m+n+1)+B*(a*d*m+b*c*n))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,m},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && GtQ[n,0] && (IntegerQ[n] || EqQ[m+1/2,0])
```

7: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,n<-1$

Derivation: Singly degenerate sine recurrence 1c with $p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,n<-1$, then

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx\longrightarrow$$

$$\frac{(B\,c-A\,d)\,\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^{n+1}}{f\,(n+1)\,(c^2-d^2)}+$$

$$\frac{1}{b\,(n+1)\,(c^2-d^2)}\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^{n+1}\,(A\,(a\,d\,m+b\,c\,(n+1))-B\,(a\,c\,m+b\,d\,(n+1))+b\,(B\,c-A\,d)\,(m+n+2)\,\text{Sin}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_*(A_.+B_.*sin[e_.+f_.*x_]),x_Symbol] :=
  (B*c-A*d)*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^(n+1)/(f*(n+1)*(c^2-d^2)) +
  1/(b*(n+1)*(c^2-d^2))*Int[(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^(n+1)*
    Simp[A*(a*d*m+b*c*(n+1))-B*(a*c*m+b*d*(n+1))+b*(B*c-A*d)*(m+n+2)*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,m},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && LtQ[n,-1] && (IntegerQ[n] || EqQ[m+1/2,0])
```

8. $\int \frac{(a+b\,\mathrm{Sin}[e+f\,x])^m\,(A+B\,\mathrm{Sin}[e+f\,x])}{c+d\,\mathrm{Sin}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0$

1: $\int \frac{A+B\,\mathrm{Sin}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,(c+d\,\mathrm{Sin}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{A+B\,z}{\sqrt{a+b\,z}\,(c+d\,z)}==\frac{A\,b-a\,B}{(b\,c-a\,d)\,\sqrt{a+b\,z}}+\frac{(B\,c-A\,d)\,\sqrt{a+b\,z}}{(b\,c-a\,d)\,(c+d\,z)}$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0$, then

$$\int \frac{A+B\,\mathrm{Sin}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,(c+d\,\mathrm{Sin}[e+f\,x])}\,dx \to \frac{A\,b-a\,B}{b\,c-a\,d}\int \frac{1}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}\,dx+\frac{B\,c-A\,d}{b\,c-a\,d}\int \frac{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}{c+d\,\mathrm{Sin}[e+f\,x]}\,dx$$

Program code:

```
Int[(A_.+B_.*sin[e_.+f_.*x_])/(Sqrt[a_+b_.*sin[e_.+f_.*x_]]*(c_.+d_.*sin[e_.+f_.*x_])),x_Symbol] :=
  (A*b-a*B)/(b*c-a*d)*Int[1/Sqrt[a+b*Sin[e+f*x]],x] +
  (B*c-A*d)/(b*c-a*d)*Int[Sqrt[a+b*Sin[e+f*x]]/(c+d*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

2: $\int \frac{(a+b\,\mathrm{Sin}[e+f\,x])^m\,(A+B\,\mathrm{Sin}[e+f\,x])}{c+d\,\mathrm{Sin}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,m\neq -\frac{1}{2}$

Derivation: Algebraic expansion

Baisi: $\frac{A+B\,z}{c+d\,z}==\frac{B}{d}-\frac{B\,c-A\,d}{d\,(c+d\,z)}$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,m\neq -\frac{1}{2}$, then

$$\int \frac{(a+b\,\mathrm{Sin}[e+f\,x])^m\,(A+B\,\mathrm{Sin}[e+f\,x])}{c+d\,\mathrm{Sin}[e+f\,x]}\,dx \longrightarrow \frac{B}{d}\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,dx - \frac{B\,c-A\,d}{d}\int \frac{(a+b\,\mathrm{Sin}[e+f\,x])^m}{c+d\,\mathrm{Sin}[e+f\,x]}\,dx$$

– Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(A_.+B_.*sin[e_.+f_.*x_])/(c_.+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  B/d*Int[(a+b*Sin[e+f*x])^m,x] - (B*c-A*d)/d*Int[(a+b*Sin[e+f*x])^m/(c+d*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,A,B,m},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && NeQ[m+1/2,0]
```

9: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,(A+B\,\mathrm{Sin}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\wedge a^2-b^2==0\wedge c^2-d^2\neq 0$

Derivation: Algebraic expansion

Baisi: $A+B\,z==\frac{A\,b-a\,B}{b}+\frac{B\,(a+b\,z)}{b}$

Rule: If $b\,c-a\,d\neq 0\wedge a^2-b^2==0\wedge c^2-d^2\neq 0$, then

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,(A+B\,\mathrm{Sin}[e+f\,x])\,dx \longrightarrow$$
$$\frac{A\,b-a\,B}{b}\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx+\frac{B}{b}\int (a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$$

– Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_.+d_.*sin[e_.+f_.*x_])^n_*(A_.+B_.*sin[e_.+f_.*x_]),x_Symbol] :=
  (A*b-a*B)/b*Int[(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n,x] +
  B/b*Int[(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,A,B,m,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && NeQ[A*b+a*B,0]
```

6. $\int (a + b \sin[e + f x])^m (c + d \sin[e + f x])^n (A + B \sin[e + f x]) \, dx$ when $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 \neq 0$

1. $\int (a + b \sin[e + f x])^m (c + d \sin[e + f x])^n (A + B \sin[e + f x]) \, dx$ when $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 \neq 0 \land m > 1$

1. $\int (a + b \sin[e + f x])^m (c + d \sin[e + f x])^n (A + B \sin[e + f x]) \, dx$ when $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 \neq 0 \land m > 1 \land n < -1$

1: $\int (a + b \sin[e + f x])^2 (c + d \sin[e + f x])^n (A + B \sin[e + f x]) \, dx$ when $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 \neq 0 \land n < -1$

Derivation: Nondegenerate sine recurrence 1a with $A \to a A$, $B \to A b + a B$, $C \to b B$, $m \to m - 1$, $p \to 0$

Rule: If $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 \neq 0 \land n < -1$, then

$$\int (a + b \sin[e + f x])^2 (c + d \sin[e + f x])^n (A + B \sin[e + f x]) \, dx \longrightarrow$$

$$\frac{(B c - A d) (b c - a d)^2 \cos[e + f x] (c + d \sin[e + f x])^{n+1}}{f d^2 (n + 1) (c^2 - d^2)} - \frac{1}{d^2 (n + 1) (c^2 - d^2)} \int (c + d \sin[e + f x])^{n+1} \cdot$$
$$\Big(d (n + 1) \left(B (b c - a d)^2 - A d (a^2 c + b^2 c - 2 a b d)\right) -$$
$$\left((B c - A d) (a^2 d^2 (n + 2) + b^2 (c^2 + d^2 (n + 1))) + 2 a b d (A c d (n + 2) - B (c^2 + d^2 (n + 1)))\right) \sin[e + f x] -$$
$$b^2 B d (n + 1) (c^2 - d^2) \sin[e + f x]^2\Big) \, dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^2*(c_.+d_.*sin[e_.+f_.*x_])^n_*(A_.+B_.*sin[e_.+f_.*x_]),x_Symbol] :=
  (B*c-A*d)*(b*c-a*d)^2*Cos[e+f*x]*(c+d*Sin[e+f*x])^(n+1)/(f*d^2*(n+1)*(c^2-d^2)) -
  1/(d^2*(n+1)*(c^2-d^2))*Int[(c+d*Sin[e+f*x])^(n+1)*
    Simp[d*(n+1)*(B*(b*c-a*d)^2-A*d*(a^2*c+b^2*c-2*a*b*d))-
      ((B*c-A*d)*(a^2*d^2*(n+2)+b^2*(c^2+d^2*(n+1)))+2*a*b*d*(A*c*d*(n+2)-B*(c^2+d^2*(n+1))))*Sin[e+f*x]-
      b^2*B*d*(n+1)*(c^2-d^2)*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && LtQ[n,-1]
```

2: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m>1\,\wedge\,n<-1$

- Derivation: Nondegenerate sine recurrence 1a with $A\to a\,A,\ B\to A\,b+a\,B,\ C\to b\,B,\ m\to m-1,\ p\to 0$
- Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m>1\,\wedge\,n<-1$, then

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx\,\longrightarrow$$
$$-\frac{(b\,c-a\,d)\,(B\,c-A\,d)\,\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^{m-1}\,(c+d\,\text{Sin}[e+f\,x])^{n+1}}{d\,f\,(n+1)\,(c^2-d^2)}+$$
$$\frac{1}{d\,(n+1)\,(c^2-d^2)}\int (a+b\,\text{Sin}[e+f\,x])^{m-2}\,(c+d\,\text{Sin}[e+f\,x])^{n+1}\cdot$$
$$\big(b\,(b\,c-a\,d)\,(B\,c-A\,d)\,(m-1)+a\,d\,(a\,A\,c+b\,B\,c-(A\,b+a\,B)\,d)\,(n+1)+$$
$$\big(b\,\big(b\,d\,(B\,c-A\,d)+a\,\big(A\,c\,d+B\,\big(c^2-2\,d^2\big)\big)\big)\,(n+1)-a\,(b\,c-a\,d)\,(B\,c-A\,d)\,(n+2)\big)\,\text{Sin}[e+f\,x]+$$
$$b\,\big(d\,(A\,b\,c+a\,B\,c-a\,A\,d)\,(m+n+1)-b\,B\,\big(c^2\,m+d^2\,(n+1)\big)\big)\,\text{Sin}[e+f\,x]^2\big)\,dx$$

- Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_*(A_.+B_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -(b*c-a*d)*(B*c-A*d)*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^(n+1)/(d*f*(n+1)*(c^2-d^2)) +
  1/(d*(n+1)*(c^2-d^2))*Int[(a+b*Sin[e+f*x])^(m-2)*(c+d*Sin[e+f*x])^(n+1)*
    Simp[b*(b*c-a*d)*(B*c-A*d)*(m-1)+a*d*(a*A*c+b*B*c-(A*b+a*B)*d)*(n+1)+
      (b*(b*d*(B*c-A*d)+a*(A*c*d+B*(c^2-2*d^2)))*(n+1)-a*(b*c-a*d)*(B*c-A*d)*(n+2))*Sin[e+f*x]+
      b*(d*(A*b*c+a*B*c-a*A*d)*(m+n+1)-b*B*(c^2*m+d^2*(n+1)))*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && GtQ[m,1] && LtQ[n,-1]
```

2: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m>1\,\wedge\,n\nless-1$

- Derivation: Nondegenerate sine recurrence 1b with $A\to a\,A,\ B\to A\,b+a\,B,\ C\to b\,B,\ m\to m-1,\ p\to 0$
- Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m>1\,\wedge\,n\nless-1$, then

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx \longrightarrow$$
$$-\frac{b\,B\,\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^{m-1}\,(c+d\,\text{Sin}[e+f\,x])^{n+1}}{d\,f\,(m+n+1)}+$$
$$\frac{1}{d\,(m+n+1)}\int (a+b\,\text{Sin}[e+f\,x])^{m-2}\,(c+d\,\text{Sin}[e+f\,x])^n\cdot$$
$$\left(a^2\,A\,d\,(m+n+1)+b\,B\,(b\,c\,(m-1)+a\,d\,(n+1))+\right.$$
$$(a\,d\,(2\,A\,b+a\,B)\,(m+n+1)-b\,B\,(a\,c-b\,d\,(m+n)))\,\text{Sin}[e+f\,x]+$$
$$\left.b\,(A\,b\,d\,(m+n+1)-B\,(b\,c\,m-a\,d\,(2\,m+n)))\,\text{Sin}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_*(A_.+B_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -b*B*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^(n+1)/(d*f*(m+n+1)) +
  1/(d*(m+n+1))*Int[(a+b*Sin[e+f*x])^(m-2)*(c+d*Sin[e+f*x])^n*
    Simp[a^2*A*d*(m+n+1)+b*B*(b*c*(m-1)+a*d*(n+1))+
      (a*d*(2*A*b+a*B)*(m+n+1)-b*B*(a*c-b*d*(m+n)))*Sin[e+f*x]+
      b*(A*b*d*(m+n+1)-B*(b*c*m-a*d*(2*m+n)))*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && GtQ[m,1] && Not[IGtQ[n,1] &&
  (Not[IntegerQ[m]] || EqQ[a,0] && NeQ[c,0])]
```

2. $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m<-1$

1. $\int \frac{\sqrt{c+d\,\text{Sin}[e+f\,x]}\,(A+B\,\text{Sin}[e+f\,x])}{(a+b\,\text{Sin}[e+f\,x])^{3/2}}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0$

1: $\int \frac{\sqrt{c+d\,\text{Sin}[e+f\,x]}\,(A+B\,\text{Sin}[e+f\,x])}{(b\,\text{Sin}[e+f\,x])^{3/2}}\,dx$ when $c^2-d^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{(A+B\,z)\,\sqrt{c+d\,z}}{(b\,z)^{3/2}} == \frac{B\,d\,\sqrt{b\,z}}{b^2\,\sqrt{c+d\,z}} + \frac{A\,c+(B\,c+A\,d)\,z}{(b\,z)^{3/2}\,\sqrt{c+d\,z}}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,c^2-d^2\neq 0$, then

$$\int \frac{\sqrt{c + d \operatorname{Sin}[e + f x]} \; (A + B \operatorname{Sin}[e + f x])}{(b \operatorname{Sin}[e + f x])^{3/2}} \, dx \to \frac{B d}{b^2} \int \frac{\sqrt{b \operatorname{Sin}[e + f x]}}{\sqrt{c + d \operatorname{Sin}[e + f x]}} \, dx + \int \frac{A c + (B c + A d) \operatorname{Sin}[e + f x]}{(b \operatorname{Sin}[e + f x])^{3/2} \sqrt{c + d \operatorname{Sin}[e + f x]}} \, dx$$

–

Program code:

```
Int[Sqrt[c_+d_.*sin[e_.+f_.*x_]]*(A_.+B_.*sin[e_.+f_.*x_])/(b_.*sin[e_.+f_.*x_])^(3/2),x_Symbol] :=
  B*d/b^2*Int[Sqrt[b*Sin[e+f*x]]/Sqrt[c+d*Sin[e+f*x]],x] +
  Int[(A*c+(B*c+A*d)*Sin[e+f*x])/((b*Sin[e+f*x])^(3/2)*Sqrt[c+d*Sin[e+f*x]]),x] /;
FreeQ[{b,c,d,e,f,A,B},x] && NeQ[c^2-d^2,0]
```

2: $\int \frac{\sqrt{c + d \operatorname{Sin}[e + f x]} \; (A + B \operatorname{Sin}[e + f x])}{(a + b \operatorname{Sin}[e + f x])^{3/2}} \, dx$ when $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{A+B z}{a+b z} == \frac{B}{b} + \frac{A b-a B}{b (a+b z)}$

–

Rule: If $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 \neq 0$, then

$$\int \frac{\sqrt{c + d \operatorname{Sin}[e + f x]} \; (A + B \operatorname{Sin}[e + f x])}{(a + b \operatorname{Sin}[e + f x])^{3/2}} \, dx \to \frac{B}{b} \int \frac{\sqrt{c + d \operatorname{Sin}[e + f x]}}{\sqrt{a + b \operatorname{Sin}[e + f x]}} \, dx + \frac{A b - a B}{b} \int \frac{\sqrt{c + d \operatorname{Sin}[e + f x]}}{(a + b \operatorname{Sin}[e + f x])^{3/2}} \, dx$$

Program code:

```
Int[Sqrt[c_.+d_.*sin[e_.+f_.*x_]]*(A_.+B_.*sin[e_.+f_.*x_])/(a_+b_.*sin[e_.+f_.*x_])^(3/2),x_Symbol] :=
  B/b*Int[Sqrt[c+d*Sin[e+f*x]]/Sqrt[a+b*Sin[e+f*x]],x] +
  (A*b-a*B)/b*Int[Sqrt[c+d*Sin[e+f*x]]/(a+b*Sin[e+f*x])^(3/2),x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

2. $\int \frac{A+B\,\text{Sin}[e+f\,x]}{(a+b\,\text{Sin}[e+f\,x])^{3/2}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0$

1: $\int \frac{A+B\,\text{Sin}[e+f\,x]}{(a+b\,\text{Sin}[e+f\,x])^{3/2}\,\sqrt{d\,\text{Sin}[e+f\,x]}}\,dx$ when $a^2-b^2\neq 0$

- Derivation: Nondegenerate sine recurrence 1a with $c\to 0$, $C\to 0$, $m\to -\frac{3}{2}$, $n\to -\frac{1}{2}$, $p\to 0$

- Rule: If $a^2-b^2\neq 0$, then

$$\int \frac{A+B\,\text{Sin}[e+f\,x]}{(a+b\,\text{Sin}[e+f\,x])^{3/2}\,\sqrt{d\,\text{Sin}[e+f\,x]}}\,dx\;\to\;\frac{2\,(A\,b-a\,B)\,\text{Cos}[e+f\,x]}{f\,(a^2-b^2)\,\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{d\,\text{Sin}[e+f\,x]}}+\frac{d}{(a^2-b^2)}\int \frac{A\,b-a\,B+(a\,A-b\,B)\,\text{Sin}[e+f\,x]}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,(d\,\text{Sin}[e+f\,x])^{3/2}}\,dx$$

- Program code:

```
Int[(A_.+B_.*sin[e_.+f_.*x_])/((a_+b_.*sin[e_.+f_.*x_])^(3/2)*Sqrt[d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  2*(A*b-a*B)*Cos[e+f*x]/(f*(a^2-b^2)*Sqrt[a+b*Sin[e+f*x]]*Sqrt[d*Sin[e+f*x]]) +
  d/(a^2-b^2)*Int[(A*b-a*B+(a*A-b*B)*Sin[e+f*x])/(Sqrt[a+b*Sin[e+f*x]]*(d*Sin[e+f*x])^(3/2)),x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[a^2-b^2,0]
```

2. $\int \frac{A+B\,\text{Sin}[e+f\,x]}{(a+b\,\text{Sin}[e+f\,x])^{3/2}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0$

1. $\int \frac{A+B\,\text{Sin}[e+f\,x]}{(a+b\,\text{Sin}[e+f\,x])^{3/2}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0\,\land\,A=B$

1. $\int \frac{A+B\,\text{Sin}[e+f\,x]}{(b\,\text{Sin}[e+f\,x])^{3/2}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$ when $c^2-d^2\neq 0\,\land\,A=B$

1: $\int \frac{A+B\,\text{Sin}[e+f\,x]}{(b\,\text{Sin}[e+f\,x])^{3/2}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$ when $c^2-d^2\neq 0\,\land\,A=B\,\land\,\frac{c+d}{b}>0$

- Rule: If $c^2-d^2\neq 0\,\land\,A=B\,\land\,\frac{c+d}{b}>0$, then

$$\int \frac{A+B\,\text{Sin}[e+f\,x]}{(b\,\text{Sin}[e+f\,x])^{3/2}\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx \to$$

$$-\frac{2\,A\,(c-d)\,\text{Tan}[e+f\,x]}{f\,b\,c^2}\sqrt{\frac{c+d}{b}}\sqrt{\frac{c\,(1+\text{Csc}[e+f\,x])}{c-d}}\sqrt{\frac{c\,(1-\text{Csc}[e+f\,x])}{c+d}}\,\text{EllipticE}\left[\text{ArcSin}\left[\frac{\sqrt{c+d\,\text{Sin}[e+f\,x]}}{\sqrt{b\,\text{Sin}[e+f\,x]}}\Big/\sqrt{\frac{c+d}{b}}\right],-\frac{c+d}{c-d}\right]$$

Program code:

```
Int[(A_+B_.*sin[e_.+f_.*x_])/((b_.*sin[e_.+f_.*x_])^(3/2)*Sqrt[c_+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  -2*A*(c-d)*Tan[e+f*x]/(f*b*c^2)*Rt[(c+d)/b,2]*Sqrt[c*(1+Csc[e+f*x])/(c-d)]*Sqrt[c*(1-Csc[e+f*x])/(c+d)]*
    EllipticE[ArcSin[Sqrt[c+d*Sin[e+f*x]]/Sqrt[b*Sin[e+f*x]]/Rt[(c+d)/b,2]],-(c+d)/(c-d)] /;
FreeQ[{b,c,d,e,f,A,B},x] && NeQ[c^2-d^2,0] && EqQ[A,B] && PosQ[(c+d)/b]
```

2: $\int \frac{A+B\,\text{Sin}[e+f\,x]}{(b\,\text{Sin}[e+f\,x])^{3/2}\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$ when $c^2-d^2\neq 0\land A==B\land \frac{c+d}{b}\ngtr 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{-F[x]}}{\sqrt{F[x]}} == 0$

Rule: If $c^2-d^2\neq 0\land A==B\land \frac{c+d}{b}\ngtr 0$, then

$$\int \frac{A+B\,\text{Sin}[e+f\,x]}{(b\,\text{Sin}[e+f\,x])^{3/2}\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx \to -\frac{\sqrt{-b\,\text{Sin}[e+f\,x]}}{\sqrt{b\,\text{Sin}[e+f\,x]}}\int \frac{A+B\,\text{Sin}[e+f\,x]}{(-b\,\text{Sin}[e+f\,x])^{3/2}\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$$

Program code:

```
Int[(A_+B_.*sin[e_.+f_.*x_])/((b_.*sin[e_.+f_.*x_])^(3/2)*Sqrt[c_+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  -Sqrt[-b*Sin[e+f*x]]/Sqrt[b*Sin[e+f*x]]*Int[(A+B*Sin[e+f*x])/((-b*Sin[e+f*x])^(3/2)*Sqrt[c+d*Sin[e+f*x]]),x] /;
FreeQ[{b,c,d,e,f,A,B},x] && NeQ[c^2-d^2,0] && EqQ[A,B] && NegQ[(c+d)/b]
```

2. $\int \frac{A + B \sin[e + f x]}{(a + b \sin[e + f x])^{3/2} \sqrt{c + d \sin[e + f x]}} \, dx$ when $b c - a d \neq 0 \wedge a^2 - b^2 \neq 0 \wedge c^2 - d^2 \neq 0 \wedge A == B$

1: $\int \frac{A + B \sin[e + f x]}{(a + b \sin[e + f x])^{3/2} \sqrt{c + d \sin[e + f x]}} \, dx$ when $b c - a d \neq 0 \wedge a^2 - b^2 \neq 0 \wedge c^2 - d^2 \neq 0 \wedge A == B \wedge \frac{a+b}{c+d} > 0$

Rule: If $b c - a d \neq 0 \wedge a^2 - b^2 \neq 0 \wedge c^2 - d^2 \neq 0 \wedge A == B \wedge \frac{a+b}{c+d} > 0$, then

$$\int \frac{A + B \sin[e + f x]}{(a + b \sin[e + f x])^{3/2} \sqrt{c + d \sin[e + f x]}} \, dx \to$$

$$-\frac{2 A (c - d) (a + b \sin[e + f x])}{f (b c - a d)^2 \sqrt{\frac{a+b}{c+d}} \cos[e + f x]} \sqrt{\frac{(b c - a d) (1 + \sin[e + f x])}{(c - d) (a + b \sin[e + f x])}}$$

$$\sqrt{-\frac{(b c - a d) (1 - \sin[e + f x])}{(c + d) (a + b \sin[e + f x])}} \, \text{EllipticE}\left[\text{ArcSin}\left[\sqrt{\frac{a + b}{c + d}} \frac{\sqrt{c + d \sin[e + f x]}}{\sqrt{a + b \sin[e + f x]}}\right], \frac{(a - b) (c + d)}{(a + b) (c - d)}\right]$$

Program code:

```
Int[(A_+B_.*sin[e_.+f_.*x_])/((a_+b_.*sin[e_.+f_.*x_])^(3/2)*Sqrt[c_+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  -2*A*(c-d)*(a+b*Sin[e+f*x])/(f*(b*c-a*d)^2*Rt[(a+b)/(c+d),2]*Cos[e+f*x])*
    Sqrt[(b*c-a*d)*(1+Sin[e+f*x])/((c-d)*(a+b*Sin[e+f*x]))]*
    Sqrt[-(b*c-a*d)*(1-Sin[e+f*x])/((c+d)*(a+b*Sin[e+f*x]))]*
    EllipticE[ArcSin[Rt[(a+b)/(c+d),2]*Sqrt[c+d*Sin[e+f*x]]/Sqrt[a+b*Sin[e+f*x]]],(a-b)*(c+d)/((a+b)*(c-d))] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && EqQ[A,B] && PosQ[(a+b)/(c+d)]
```

2: $\int \frac{A + B \sin[e + f x]}{(a + b \sin[e + f x])^{3/2} \sqrt{c + d \sin[e + f x]}} \, dx$ when $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 \neq 0 \land A == B \land \frac{a+b}{c+d} \ngtr 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{-F[x]}}{\sqrt{F[x]}} == 0$

Rule: If $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 \neq 0 \land A == B \land \frac{a+b}{c+d} \ngtr 0$, then

$$\int \frac{A + B \sin[e + f x]}{(a + b \sin[e + f x])^{3/2} \sqrt{c + d \sin[e + f x]}} \, dx \to \frac{\sqrt{-c - d \sin[e + f x]}}{\sqrt{c + d \sin[e + f x]}} \int \frac{A + B \sin[e + f x]}{(a + b \sin[e + f x])^{3/2} \sqrt{-c - d \sin[e + f x]}} \, dx$$

Program code:

```
Int[(A_+B_.*sin[e_.+f_.*x_])/((a_+b_.*sin[e_.+f_.*x_])^(3/2)*Sqrt[c_+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  Sqrt[-c-d*Sin[e+f*x]]/Sqrt[c+d*Sin[e+f*x]]*Int[(A+B*Sin[e+f*x])/((a+b*Sin[e+f*x])^(3/2)*Sqrt[-c-d*Sin[e+f*x]]),x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && EqQ[A,B] && NegQ[(a+b)/(c+d)]
```

2: $\int \frac{A + B \sin[e + f x]}{(a + b \sin[e + f x])^{3/2} \sqrt{c + d \sin[e + f x]}} \, dx$ when $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 \neq 0 \land A \neq B$

Derivation: Algebraic expansion

Basis: $\frac{A+B z}{(a+b z)^{3/2}} == \frac{A-B}{(a-b) \sqrt{a+b z}} - \frac{(A b-a B) (1+z)}{(a-b) (a+b z)^{3/2}}$

Rule: If $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 \neq 0 \land A \neq B$, then

$$\int \frac{A + B \sin[e + f x]}{(a + b \sin[e + f x])^{3/2} \sqrt{c + d \sin[e + f x]}} \, dx \to$$

$$\frac{A-B}{a-b}\int\frac{1}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx-\frac{A\,b-a\,B}{a-b}\int\frac{1+\text{Sin}[e+f\,x]}{(a+b\,\text{Sin}[e+f\,x])^{3/2}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$$

Program code:

```
Int[(A_.+B_.*sin[e_.+f_.*x_])/((a_.+b_.*sin[e_.+f_.*x_])^(3/2)*Sqrt[c_+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  (A-B)/(a-b)*Int[1/(Sqrt[a+b*Sin[e+f*x]]*Sqrt[c+d*Sin[e+f*x]]),x] -
  (A*b-a*B)/(a-b)*Int[(1+Sin[e+f*x])/((a+b*Sin[e+f*x])^(3/2)*Sqrt[c+d*Sin[e+f*x]]),x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && NeQ[A,B]
```

3. $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0\,\land\,m<-1$

1: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0\,\land\,m<-1\,\land\,n>0$

Derivation: Nondegenerate sine recurrence 1a with $C\to 0,\ p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0\,\land\,m<-1\,\land\,n>0$, then

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx\longrightarrow$$
$$\frac{(B\,a-A\,b)\,\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^{m+1}\,(c+d\,\text{Sin}[e+f\,x])^n}{f\,(m+1)\,(a^2-b^2)}+$$
$$\frac{1}{(m+1)\,(a^2-b^2)}\int (a+b\,\text{Sin}[e+f\,x])^{m+1}\,(c+d\,\text{Sin}[e+f\,x])^{n-1}\cdot$$
$$(c\,(a\,A-b\,B)\,(m+1)+d\,n\,(A\,b-a\,B)+(d\,(a\,A-b\,B)\,(m+1)-c\,(A\,b-a\,B)\,(m+2))\,\text{Sin}[e+f\,x]-d\,(A\,b-a\,B)\,(m+n+2)\,\text{Sin}[e+f\,x]^2)\,dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_*(A_.+B_.*sin[e_.+f_.*x_]),x_Symbol] :=
  (B*a-A*b)*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^n/(f*(m+1)*(a^2-b^2)) +
  1/((m+1)*(a^2-b^2))*Int[(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^(n-1)*
    Simp[c*(a*A-b*B)*(m+1)+d*n*(A*b-a*B)+(d*(a*A-b*B)*(m+1)-c*(A*b-a*B)*(m+2))*Sin[e+f*x]-d*(A*b-a*B)*(m+n+2)*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && LtQ[m,-1] && GtQ[n,0]
```

2: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,(A+B\,\mathrm{Sin}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0\,\land\,m<-1\,\land\,n\not>0$

Derivation: Nondegenerate sine recurrence 1c with $C\to 0,\ p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0\,\land\,m<-1\,\land\,n\not>0$, then

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,(A+B\,\mathrm{Sin}[e+f\,x])\,dx\longrightarrow$$
$$-\frac{b\,(A\,b-a\,B)\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Sin}[e+f\,x])^{n+1}}{f\,(m+1)\,(b\,c-a\,d)\,(a^2-b^2)}+$$
$$\frac{1}{(m+1)\,(b\,c-a\,d)\,(a^2-b^2)}\int (a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Sin}[e+f\,x])^n\cdot$$
$$\left((a\,A-b\,B)\,(b\,c-a\,d)\,(m+1)+b\,d\,(A\,b-a\,B)\,(m+n+2)+(A\,b-a\,B)\,(a\,d\,(m+1)-b\,c\,(m+2))\,\mathrm{Sin}[e+f\,x]-b\,d\,(A\,b-a\,B)\,(m+n+3)\,\mathrm{Sin}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_*(A_.+B_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -(A*b^2-a*b*B)*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^(1+n)/(f*(m+1)*(b*c-a*d)*(a^2-b^2)) +
  1/((m+1)*(b*c-a*d)*(a^2-b^2))*Int[(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^n*
    Simp[(a*A-b*B)*(b*c-a*d)*(m+1)+b*d*(A*b-a*B)*(m+n+2)+
      (A*b-a*B)*(a*d*(m+1)-b*c*(m+2))*Sin[e+f*x]-
      b*d*(A*b-a*B)*(m+n+3)*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && RationalQ[m] && m<-1 &&
  (EqQ[a,0] && IntegerQ[m] && Not[IntegerQ[n]] || Not[IntegerQ[2*n] && LtQ[n,-1] && (IntegerQ[n] && Not[IntegerQ[m]] || EqQ[a,0])])
```

3. $\int \frac{(a+b\,\text{Sin}[e+f\,x])^m\,(A+B\,\text{Sin}[e+f\,x])}{c+d\,\text{Sin}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0$

1: $\int \frac{A+B\,\text{Sin}[e+f\,x]}{(a+b\,\text{Sin}[e+f\,x])\,(c+d\,\text{Sin}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{A+B\,z}{(a+b\,z)\,(c+d\,z)} == \frac{A\,b-a\,B}{(b\,c-a\,d)\,(a+b\,z)} + \frac{B\,c-A\,d}{(b\,c-a\,d)\,(c+d\,z)}$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0$, then

$$\int \frac{A+B\,\text{Sin}[e+f\,x]}{(a+b\,\text{Sin}[e+f\,x])\,(c+d\,\text{Sin}[e+f\,x])}\,dx \longrightarrow \frac{A\,b-a\,B}{b\,c-a\,d}\int \frac{1}{a+b\,\text{Sin}[e+f\,x]}\,dx + \frac{B\,c-A\,d}{b\,c-a\,d}\int \frac{1}{c+d\,\text{Sin}[e+f\,x]}\,dx$$

Program code:

```
Int[(A_.+B_.*sin[e_.+f_.*x_])/((a_.+b_.*sin[e_.+f_.*x_])*(c_.+d_.*sin[e_.+f_.*x_])),x_Symbol] :=
  (A*b-a*B)/(b*c-a*d)*Int[1/(a+b*Sin[e+f*x]),x] + (B*c-A*d)/(b*c-a*d)*Int[1/(c+d*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

2: $\int \frac{(a+b\,\text{Sin}[e+f\,x])^m\,(A+B\,\text{Sin}[e+f\,x])}{c+d\,\text{Sin}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{A+B\,z}{c+d\,z}==\frac{B}{d}-\frac{B\,c-A\,d}{d\,(c+d\,z)}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0$, then

$$\int \frac{(a+b\,\text{Sin}[e+f\,x])^m\,(A+B\,\text{Sin}[e+f\,x])}{c+d\,\text{Sin}[e+f\,x]}\,dx \longrightarrow \frac{B}{d}\int (a+b\,\text{Sin}[e+f\,x])^m\,dx-\frac{B\,c-A\,d}{d}\int \frac{(a+b\,\text{Sin}[e+f\,x])^m}{c+d\,\text{Sin}[e+f\,x]}\,dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_*(A_.+B_.*sin[e_.+f_.*x_])/(c_.+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  B/d*Int[(a+b*Sin[e+f*x])^m,x] - (B*c-A*d)/d*Int[(a+b*Sin[e+f*x])^m/(c+d*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,A,B,m},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

4: $\int\sqrt{a+b\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0\,\land\,n^2==\frac{1}{4}$

- Derivation: Nondegenerate sine recurrence 1b with $A\to A\,c$, $B\to B\,c+A\,d$, $C\to B\,d$, $n\to n-1$, $p\to 0$

- Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0\,\land\,n^2==\frac{1}{4}$, then

$$\int\sqrt{a+b\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x])\,dx\,\longrightarrow$$

$$-\frac{2\,B\,\text{Cos}[e+f\,x]\,\sqrt{a+b\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])^n}{f\,(2\,n+3)}+\frac{1}{2\,n+3}\int\frac{(c+d\,\text{Sin}[e+f\,x])^{n-1}}{\sqrt{a+b\,\text{Sin}[e+f\,x]}}\,\cdot$$

$$\left(a\,A\,c\,(2\,n+3)+B\,(b\,c+2\,a\,d\,n)+(B\,(a\,c+b\,d)\,(2\,n+1)+A\,(b\,c+a\,d)\,(2\,n+3))\,\text{Sin}[e+f\,x]+(A\,b\,d\,(2\,n+3)+B\,(a\,d+2\,b\,c\,n))\,\text{Sin}[e+f\,x]^2\right)\,dx$$

- Program code:

```
Int[Sqrt[a_.+b_.*sin[e_.+f_.*x_]]*(c_.+d_.*sin[e_.+f_.*x_])^n_*(A_.+B_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -2*B*Cos[e+f*x]*Sqrt[a+b*Sin[e+f*x]]*(c+d*Sin[e+f*x])^n/(f*(2*n+3)) +
  1/(2*n+3)*Int[(c+d*Sin[e+f*x])^(n-1)/Sqrt[a+b*Sin[e+f*x]]*
     Simp[a*A*c*(2*n+3)+B*(b*c+2*a*d*n)+
       (B*(a*c+b*d)*(2*n+1)+A*(b*c+a*d)*(2*n+3))*Sin[e+f*x]+
       (A*b*d*(2*n+3)+B*(a*d+2*b*c*n))*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && EqQ[n^2,1/4]
```

5. $\int \frac{A+B\,\mathrm{Sin}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0$

   1. $\int \frac{A+B\,\mathrm{Sin}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{d\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $b>0\,\wedge\,b^2-a^2>0\,\wedge\,A==B$

      1: $\int \frac{A+B\,\mathrm{Sin}[e+f\,x]}{\sqrt{\mathrm{Sin}[e+f\,x]}\,\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $b>0\,\wedge\,b^2-a^2>0\,\wedge\,A==B$

Derivation: Algebraic expansion

Basis: If $b>0\,\wedge\,b-a>0$, then $\sqrt{a+b\,z}=\sqrt{1+z}\,\sqrt{\frac{a+b\,z}{1+z}}$

Rule: If $b>0\,\wedge\,b^2-a^2>0\,\wedge\,A==B$, then

$$\int \frac{A+B\,\mathrm{Sin}[e+f\,x]}{\sqrt{\mathrm{Sin}[e+f\,x]}\,\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}\,dx\;\longrightarrow\;\frac{4\,A}{f\,\sqrt{a+b}}\,\mathrm{EllipticPi}\left[-1,\,-\mathrm{ArcSin}\left[\frac{\mathrm{Cos}[e+f\,x]}{1+\mathrm{Sin}[e+f\,x]}\right],\,-\frac{a-b}{a+b}\right]$$

Program code:

```
Int[(A_+B_.*sin[e_.+f_.*x_])/(Sqrt[sin[e_.+f_.*x_]]*Sqrt[a_+b_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  4*A/(f*Sqrt[a+b])*EllipticPi[-1,-ArcSin[Cos[e+f*x]/(1+Sin[e+f*x])],-(a-b)/(a+b)] /;
FreeQ[{a,b,e,f,A,B},x] && GtQ[b,0] && GtQ[b^2-a^2,0] && EqQ[A,B]
```

2: $\int \frac{A + B \sin[e + f x]}{\sqrt{a + b \sin[e + f x]}\ \sqrt{d \sin[e + f x]}}\, dx$ when $b > 0 \land b^2 - a^2 > 0 \land A == B$

Derivation: Piecewise constant extraction

Basis: $\partial_z \frac{\sqrt{f[z]}}{\sqrt{d\, f[z]}} == 0$

Rule: If $a^2 - b^2 \neq 0 \land A == B$, then

$$\int \frac{A + B \sin[e + f x]}{\sqrt{a + b \sin[e + f x]}\ \sqrt{d \sin[e + f x]}}\, dx \to \frac{\sqrt{\sin[e + f x]}}{\sqrt{d \sin[e + f x]}} \int \frac{A + B \sin[e + f x]}{\sqrt{\sin[e + f x]}\ \sqrt{a + b \sin[e + f x]}}\, dx$$

Program code:

```
Int[(A_+B_.*sin[e_.+f_.*x_])/(Sqrt[a_+b_.*sin[e_.+f_.*x_]]*Sqrt[d_*sin[e_.+f_.*x_]]),x_Symbol] :=
  Sqrt[Sin[e+f*x]]/Sqrt[d*Sin[e+f*x]]*Int[(A+B*Sin[e+f*x])/(Sqrt[Sin[e+f*x]]*Sqrt[a+b*Sin[e+f*x]]),x] /;
FreeQ[{a,b,e,f,d,A,B},x] && GtQ[b,0] && GtQ[b^2-a^2,0] && EqQ[A,B]
```

2: $\int \frac{A + B \sin[e + f x]}{\sqrt{a + b \sin[e + f x]} \sqrt{c + d \sin[e + f x]}} \, dx$ when $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{A+B z}{\sqrt{c+d z}} == \frac{B \sqrt{c+d z}}{d} - \frac{B c-A d}{d \sqrt{c+d z}}$

Rule: If $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 \neq 0$, then

$$\int \frac{A + B \sin[e + f x]}{\sqrt{a + b \sin[e + f x]} \sqrt{c + d \sin[e + f x]}} \, dx \to \frac{B}{d} \int \frac{\sqrt{c + d \sin[e + f x]}}{\sqrt{a + b \sin[e + f x]}} \, dx - \frac{B c - A d}{d} \int \frac{1}{\sqrt{a + b \sin[e + f x]} \sqrt{c + d \sin[e + f x]}} \, dx$$

Program code:

```
Int[(A_.+B_.*sin[e_.+f_.*x_])/(Sqrt[a_+b_.*sin[e_.+f_.*x_]]*Sqrt[c_.+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  B/d*Int[Sqrt[c+d*Sin[e+f*x]]/Sqrt[a+b*Sin[e+f*x]],x] -
  (B*c-A*d)/d*Int[1/(Sqrt[a+b*Sin[e+f*x]]*Sqrt[c+d*Sin[e+f*x]]),x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

X: $\int (a + b \sin[e + f x])^m (c + d \sin[e + f x])^n (A + B \sin[e + f x]) \, dx$ when $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 \neq 0$

Rule: If $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 \neq 0$, then

$$\int (a + b \sin[e + f x])^m (c + d \sin[e + f x])^n (A + B \sin[e + f x]) \, dx \to \int (a + b \sin[e + f x])^m (c + d \sin[e + f x])^n (A + B \sin[e + f x]) \, dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_.*(A_.+B_.*sin[e_.+f_.*x_]),x_Symbol] :=
  Unintegrable[(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n*(A+B*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,A,B,m,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

Rules for integrands of the form $(a + b \operatorname{Sin}[e + f x])^m (c + d \operatorname{Sin}[e + f x])^n (A + B \operatorname{Sin}[e + f x])^p$

x: $\int (a + b \operatorname{Sin}[e + f x])^m (c + d \operatorname{Sin}[e + f x])^n (A + B \operatorname{Sin}[e + f x])^p \, dx$ when $b c + a d = 0 \land a^2 - b^2 = 0 \land m \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b c + a d = 0 \land a^2 - b^2 = 0$, then $(a + b \operatorname{Sin}[z]) (c + d \operatorname{Sin}[z]) = a c \operatorname{Cos}[z]^2$

Rule: If $b c + a d = 0 \land a^2 - b^2 = 0 \land m \in \mathbb{Z}$, then

$$\int (a + b \operatorname{Sin}[e + f x])^m (c + d \operatorname{Sin}[e + f x])^n (A + B \operatorname{Sin}[e + f x])^p \, dx \longrightarrow a^m c^m \int \operatorname{Cos}[e + f x]^{2m} (c + d \operatorname{Sin}[e + f x])^{n-m} (A + B \operatorname{Sin}[e + f x])^p \, dx$$

Program code:

```
(* Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_*(A_.+B_.*sin[e_.+f_.*x_])^p_,x_Symbol] :=
  a^m*c^m*Int[Cos[e+f*x]^(2*m)*(c+d*Sin[e+f*x])^(n-m)*(A+B*Sin[e+f*x])^p,x] /;
FreeQ[{a,b,c,d,e,f,A,B,n,p},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IntegerQ[m] &&
  Not[IntegerQ[n] && (LtQ[m,0] && GtQ[n,0] || LtQ[0,n,m] || LtQ[m,n,0])] *)
```

```
(* Int[(a_+b_.*cos[e_.+f_.*x_])^m_*(c_+d_.*cos[e_.+f_.*x_])^n_*(A_.+B_.*cos[e_.+f_.*x_])^p_,x_Symbol] :=
  a^m*c^m*Int[Sin[e+f*x]^(2*m)*(c+d*Cos[e+f*x])^(n-m)*(A+B*Cos[e+f*x])^p,x] /;
FreeQ[{a,b,c,d,e,f,A,B,n,p},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IntegerQ[m] &&
  Not[IntegerQ[n] && (LtQ[m,0] && GtQ[n,0] || LtQ[0,n,m] || LtQ[m,n,0])] *)
```

2: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,(A+B\,\mathrm{Sin}[e+f\,x])^p\,dx$ when $b\,c+a\,d=0\,\land\,a^2-b^2=0\,\land\,m\notin\mathbb{Z}\,\land\,n\notin\mathbb{Z}\,\land\,p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $b\,c+a\,d=0\,\land\,a^2-b^2=0$, then $\partial_x \frac{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}{\mathrm{Cos}[e+f\,x]}=0$

Basis: $\mathrm{Cos}[e+f\,x]=\frac{1}{f}\,\partial_x\,\mathrm{Sin}[e+f\,x]$

Rule: If $b\,c+a\,d=0\,\land\,a^2-b^2=0\,\land\,m\notin\mathbb{Z}\,\land\,n\notin\mathbb{Z}$, then

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,(A+B\,\mathrm{Sin}[e+f\,x])^p\,dx\,\longrightarrow$$

$$\frac{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}{\mathrm{Cos}[e+f\,x]}\int \mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^{m-\frac{1}{2}}\,(c+d\,\mathrm{Sin}[e+f\,x])^{n-\frac{1}{2}}\,(A+B\,\mathrm{Sin}[e+f\,x])^p\,dx\,\longrightarrow$$

$$\frac{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}{f\,\mathrm{Cos}[e+f\,x]}\,\mathrm{Subst}\Big[\int (a+b\,x)^{m-\frac{1}{2}}\,(c+d\,x)^{n-\frac{1}{2}}\,(A+B\,x)^p\,dx,\,x,\,\mathrm{Sin}[e+f\,x]\Big]$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_+d_.*sin[e_.+f_.*x_])^n_.*(A_.+B_.*sin[e_.+f_.*x_])^p_,x_Symbol] :=
  Sqrt[a+b*Sin[e+f*x]]*Sqrt[c+d*Sin[e+f*x]]/(f*Cos[e+f*x])*
    Subst[Int[(a+b*x)^(m-1/2)*(c+d*x)^(n-1/2)*(A+B*x)^p,x],x,Sin[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,A,B,m,n,p},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0]
```

```
Int[(a_+b_.*cos[e_.+f_.*x_])^m_.*(c_+d_.*cos[e_.+f_.*x_])^n_.*(A_.+B_.*cos[e_.+f_.*x_])^p_,x_Symbol] :=
  -Sqrt[a+b*Cos[e+f*x]]*Sqrt[c+d*Cos[e+f*x]]/(f*Sin[e+f*x])*
    Subst[Int[(a+b*x)^(m-1/2)*(c+d*x)^(n-1/2)*(A+B*x)^p,x],x,Cos[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,A,B,m,n,p},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0]
```